NAG Fortran Library Routine Document

F12FDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F12FDF is an option setting routine in a suite of routines consisting of F12FDF, F12FAF, F12FBF, F12FCF and F12FEF, and may be used to supply individual optional parameters to F12FBF and F12FCF. The initialization routine F12FAF **must** have been called prior to calling F12FDF.

2 Specification

```
SUBROUTINE F12FDF (STR, ICOMM, COMM, IFAIL)
INTEGER             ICOMM(*), IFAIL
double precision    COMM(*)
CHARACTER*(*)       STR
```

3 Description

F12FDF may be used to supply values for optional parameters to F12FBF and F12FCF. It is only necessary to call F12FDF for those parameters whose values are to be different from their default values. One call to F12FDF sets one parameter value.

Each optional parameter is defined by a single character string consisting of one or more items. The items associated with a given option must be separated by spaces, or equals signs [=]. Alphabetic characters may be upper or lower case. The string

```
'Pointers = Yes'
```

is an example of a string used to set an optional parameter. For each option the string contains one or more of the following items:

(a) a mandatory keyword;

(b) a phrase that qualifies the keyword;

(c) a number that specifies an INTEGER or ***double precision*** value. Such numbers may be up to 16 contiguous characters in Fortran's I, F, E or D format.

F12FDF does not have an equivalent routine from the ARPACK package which passes options by directly setting values to scalar parameters or to specific elements of array arguments. F12FDF is intended to make the passing of options more transparent and follows the same principle as the single option setting routines in Chapter E04.

The setup routine F12FAF must be called prior to the first call to F12FDF and all calls to F12FDF must preceed the first call to F12FBF, the reverse communication iterative solver.

A complete list of optional parameters, their abbreviations, synonyms and default values is given in Section 10.

4 References

Lehoucq R B (2001) Implicitly Restarted Arnoldi Methods and Subspace Iteration *SIAM Journal on Matrix Analysis and Applications* **23** 551–562

Lehoucq R B and Scott J A (1996) An evaluation of software for computing eigenvalues of sparse nonsymmetric matrices *Preprint MCS-P547-1195* Argonne National Laboratory

Lehoucq R B and Sorensen D C (1996) Deflation Techniques for an Implicitly Restarted Arnoldi Iteration *SIAM Journal on Matrix Analysis and Applications* **17** 789–821

Lehoucq R B, Sorensen D C and Yang C (1998) *ARPACK Users' Guide: Solution of Large-Scale Eigenvalue Problems with Implicitly Restarted Arnoldi Methods* SIAM, Philidelphia

5 Parameters

1: STR – CHARACTER*(*) *Input*

On entry: a single valid option string (as described in Section 3 above and in Section 10).

2: ICOMM(∗) – INTEGER array *Communication Array*

ICOMM, on initial entry, must remain unchanged following a call to the setup routine F12FAF.

3: COMM(∗) – ***double precision*** array *Communication Array*

COMM, on initial entry, must remain unchanged following a call to the setup routine F12FAF.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The string passed in STR contains an ambiguous keyword.

IFAIL = 2

The string passed in STR contains a keyword that could not be recognized.

IFAIL = 3

The string passed in STR contains a second keyword that could not be recognized.

IFAIL = 4

The initialization routine F12FAF has not been called or a communication array has become corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

The example solves $Ax = \lambda Bx$ in **Shifted Inverse** mode, where A and B are obtained from the standard central difference discretization of the one-dimensional Laplacian operator $\frac{\partial^2 u}{\partial x^2}$ on $[0, 1]$, with zero Dirichlet boundary conditions. Data is passed to and from the reverse communication routine F12FBF using pointers to the communication array.

9.1 Program Text

```
*     F12FDF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          LICOMM, NIN, NOUT
      PARAMETER        (LICOMM=134,NIN=5,NOUT=6)
      INTEGER          MAXN, MAXNCV, LDV
      PARAMETER        (MAXN=256,MAXNCV=30,LDV=MAXN)
      INTEGER          LCOMM
      PARAMETER        (LCOMM=3*MAXN+MAXNCV*MAXNCV+8*MAXNCV+60)
      INTEGER          IMON, IPOINT
      PARAMETER        (IMON=0,IPOINT=1)
      DOUBLE PRECISION FOUR, ONE, SIX, TWO, ZERO
      PARAMETER        (FOUR=4.0D+0,ONE=1.0D+0,SIX=6.0D+0,TWO=2.0D+0,
     +                 ZERO=0.0D+0)
*     .. Local Scalars ..
      DOUBLE PRECISION H, R1, R2, SIGMA
      INTEGER          IFAIL, INFO, IREVCM, J, N, NCONV, NCV, NEV,
     +                 NITER, NSHIFT
*     .. Local Arrays ..
      DOUBLE PRECISION AD(MAXN), ADL(MAXN), ADU(MAXN), ADU2(MAXN),
     +                 COMM(LCOMM), D(MAXNCV,2), MX(MAXN), RESID(MAXN),
     +                 V(LDV,MAXNCV), X(MAXN)
      INTEGER          ICOMM(LICOMM), IPIV(MAXN)
*     .. External Functions ..
      DOUBLE PRECISION DNRM2
      EXTERNAL         DNRM2
*     .. External Subroutines ..
      EXTERNAL         DCOPY, DGTTRF, DGTTRS, F12FAF, F12FBF, F12FCF,
     +                 F12FDF, F12FEF, MV
*     .. Intrinsic Functions ..
      INTRINSIC        DBLE
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F12FDF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NEV, NCV
      IF (N.LT.1 .OR. N.GT.MAXN) THEN
         WRITE (NOUT,99999) 'N is out of range: N = ', N
      ELSE IF (NCV.GT.MAXNCV) THEN
         WRITE (NOUT,99999) 'NCV is out of range: NCV = ', NCV
      ELSE
         IFAIL = 0
         CALL F12FAF(N,NEV,NCV,ICOMM,LICOMM,COMM,LCOMM,IFAIL)
*     We are solving a generalized problem
         CALL F12FDF('GENERALIZED',ICOMM,COMM,IFAIL)
*     Indicate that we are using the shift and invert mode.
         CALL F12FDF('SHIFTED INVERSE',ICOMM,COMM,IFAIL)
         IF (IPOINT.EQ.1) THEN
*        Use pointers to Workspace in calculating matrix vector products
*        rather than interfacing through the array X
            CALL F12FDF('POINTERS=YES',ICOMM,COMM,IFAIL)
         END IF
*
         H = ONE/DBLE(N+1)
         R1 = (FOUR/SIX)*H
         R2 = (ONE/SIX)*H
         SIGMA = ZERO
         DO 20 J = 1, N
            AD(J) = TWO/H - SIGMA*R1
```

```
            ADL(J) = -ONE/H - SIGMA*R2
   20    CONTINUE
         CALL DCOPY(N,ADL,1,ADU,1)
         CALL DGTTRF(N,ADL,AD,ADU,ADU2,IPIV,INFO)
*
         IREVCM = 0
         IFAIL = -1
   40    CONTINUE
         CALL F12FBF(IREVCM,RESID,V,LDV,X,MX,NSHIFT,COMM,ICOMM,IFAIL)
         IF (IREVCM.NE.5) THEN
            IF (IREVCM.EQ.-1) THEN
*           Perform  y <--- OP*x = inv[A-SIGMA*M]*M*x.
               IF (IPOINT.EQ.0) THEN
                  CALL MV(N,X,MX)
                  CALL DCOPY(N,MX,1,X,1)
                  CALL DGTTRS('N',N,1,ADL,AD,ADU,ADU2,IPIV,X,N,INFO)
               ELSE
                  CALL MV(N,COMM(ICOMM(1)),COMM(ICOMM(2)))
                  CALL DGTTRS('N',N,1,ADL,AD,ADU,ADU2,IPIV,COMM(ICOMM(2)
     +                        ),N,INFO)
               END IF
            ELSE IF (IREVCM.EQ.1) THEN
*           Perform y <-- OP*x = inv[A-sigma*M]*M*x;
*           M*x has been saved in COMM(ICOMM(3)) or MX.
               IF (IPOINT.EQ.0) THEN
                  CALL DCOPY(N,MX,1,X,1)
                  CALL DGTTRS('N',N,1,ADL,AD,ADU,ADU2,IPIV,X,N,INFO)
               ELSE
                  CALL DCOPY(N,COMM(ICOMM(3)),1,COMM(ICOMM(2)),1)
                  CALL DGTTRS('N',N,1,ADL,AD,ADU,ADU2,IPIV,COMM(ICOMM(2)
     +                        ),N,INFO)
               END IF
            ELSE IF (IREVCM.EQ.2) THEN
*           Perform  y <--- M*x.
               IF (IPOINT.EQ.0) THEN
                  CALL MV(N,X,MX)
               ELSE
                  CALL MV(N,COMM(ICOMM(1)),COMM(ICOMM(2)))
               END IF
            ELSE IF (IREVCM.EQ.4 .AND. IMON.NE.0) THEN
*           Output monitoring information
               CALL F12FEF(NITER,NCONV,D,D(1,2),ICOMM,COMM)
               WRITE (6,99998) NITER, NCONV, DNRM2(NEV,D(1,2),1)
            END IF
            GO TO 40
         END IF
         IF (IFAIL.EQ.0) THEN
*        Post-Process using F12FCF to compute eigenvalues/values.
            CALL F12FCF(NCONV,D,V,LDV,SIGMA,RESID,V,LDV,COMM,ICOMM,
     +                  IFAIL)
            WRITE (NOUT,99996) NCONV, SIGMA
            DO 60 J = 1, NCONV
               WRITE (NOUT,99995) J, D(J,1)
   60       CONTINUE
         ELSE
            WRITE (NOUT,99997) IFAIL
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,'Iteration',1X,I3,', No. converged =',1X,I3,', norm o',
     +       'f estimates =',E16.8)
99997 FORMAT (1X,' NAG Routine F12FBF Returned with IFAIL = ',I6)
99996 FORMAT (1X,/' The ',I4,' Ritz values of closest to ',F8.4,' are:',
     +       /)
99995 FORMAT (1X,I8,5X,F12.4)
      END
*
      SUBROUTINE MV(N,V,W)
*     .. Parameters ..
```

```
      DOUBLE PRECISION ONE, FOUR, SIX
      PARAMETER     (ONE=1.0D+0,FOUR=4.0D+0,SIX=6.0D+0)
*     .. Scalar Arguments ..
      INTEGER       N
*     .. Array Arguments ..
      DOUBLE PRECISION V(N), W(N)
*     .. Local Scalars ..
      DOUBLE PRECISION H
      INTEGER       J
*     .. External Subroutines ..
      EXTERNAL      DSCAL
*     .. Intrinsic Functions ..
      INTRINSIC     DBLE
*     .. Executable Statements ..
      H = ONE/(DBLE(N+1)*SIX)
      W(1) = FOUR*V(1) + V(2)
      DO 20 J = 2, N - 1
         W(J) = V(J-1) + FOUR*V(J) + V(J+1)
   20 CONTINUE
      J = N
      W(J) = V(J-1) + FOUR*V(J)
      CALL DSCAL(N,H,W,1)
      RETURN
      END
```

9.2 Program Data

```
F12FDF Example Program Data
 100 4 10 : Values for N NEV and NCV
```

9.3 Program Results

```
 F12FDF Example Program Results

 The    4 Ritz values of closest to   0.0000 are:

        1           9.8704
        2          39.4912
        3          88.8909
        4         158.1175
```

10 Optional Parameters

Several optional parameters for the computational routines F12FBF and F12FCF define choices in the problem specification or the algorithm logic. In order to reduce the number of formal parameters of F12FBF and F12FCF these optional parameters have associated *default values* that are appropriate for most problems. Therefore, the user need only specify those optional parameters whose values are to be different from their default values.

The remainder of this section can be skipped by users who wish to use the default values for *all* optional parameters. A complete list of optional parameters and their default values is given in Section 10.1.

Optional parameters may be specified by calling F12FDF prior to a call to F12FBF, but after a call to F12FAF. One call is necessary for each optional parameter.

All optional parameters not specified by the user are set to their default values. Optional parameters specified by the user are unaltered by F12FBF and F12FCF (unless they define invalid values) and so remain in effect for subsequent calls unless altered by the user.

10.1 Optional parameter checklist and default values

The following list gives the valid options. For each option, we give the keyword, any essential optional qualifiers and the default value. A definition for each option can be found in Section 10.2. The minimum abbreviation of each keyword is underlined. The qualifier may be omitted. The letters i and r denote INTEGER and ***double precision*** values required with certain options. The number ϵ is a generic notation for ***machine precision*** (see X02AJF).

Optional Parameters	Default Values
Advisory	Default = the value returned by X04ABF
Both Ends	
Buckling	
Cayley	
Defaults	
Exact Shifts	Default = **Exact Shifts**
Generalized	
Initial Residual	
Iteration Limit	Default = 300
Largest Algebraic	
Largest Magnitude	Default = **Largest Magnitude**
List	Default = **Nolist**
Monitoring	Default = -1
Nolist	
Pointers	Default = No
Print Level	Default = 0
Random Residual	Default = **Random Residual**
Regular	Default = **Regular**
Regular Inverse	
Shifted Inverse	
Smallest Algebraic	
Smallest Magnitude	
Standard	Default = **Standard**
Supplied Shifts	
Tolerance	Default = ϵ
Vectors	Default = Schur

10.2 Description of the Optional Parameters

Advisory i Default = the value returned by X04ABF

The output channel for advisory messages.

Defaults

This special keyword may be used to reset all optional parameters to their default values.

Exact Shifts Default = **Exact Shifts**
Supplied Shifts

During the Lanczos iterative process, shifts are applied internally as part of the implicit restarting scheme. The shift strategy used by default and selected by the option **Exact Shifts** is strongly recommended over the alternative option **Supplied Shifts** (see Lehoucq *et al.* (1998) for details of shift strategies).

If **Exact Shifts** are used then these are computed internally by the algorithm in the implicit restarting scheme.

If **Supplied Shifts** are used then, during the Lanczos iterative process, you must supply shifts through array arguments of F12FBF; this option should only be used by experienced users since this requires some algorithmic knowledge and because more operations are usually required than for the implicit shift scheme.

If **Supplied Shifts** are used then, during the Lanczos iterative process, you must supply shifts through array arguments of F12FBF when F12FBF returns with IREVCM = 3; the real and imaginary parts of the shifts are returned in X and MX respectively (or in COMM when the option **Pointers** = Yes is set). This option should only be used by experienced users since this requires some algorithmic knowledge and because more operations are usually required than for the implicit shift scheme. Details on the use of explicit shifts and further references on shift strategies are available in Lehoucq *et al.* (1998).

Iteration Limit i Default $= 300$

The limit on the number of Lanczos iterations that can be performed before F12FBF exits. If not all requested eigenvalues have converged to within **Tolerance** and the number of Lanczos iterations has reached this limit then F12FBF exits with a error; F12FCF can still be called subsequently to return the number of converged eigenvalues, the converged eigenvalues and, if requested, the corresponding eigenvectors.

Largest Magnitude Default $=$ **Largest Magnitude**
Largest Algebraic
Smallest Magnitude
Smallest Algebraic
Both Ends

The Lanczos iterative method converges on a number of eigenvalues with given properties. The default is for F12FBF to compute the eigenvalues of largest magnitude using option **Largest Magnitude**. Alternatively, eigenvalues may be chosen which have **Largest Algebraic** part **Smallest Magnitude**, or **Smallest Algebraic** part; or eigenvalues which are from **Both Ends** of the algebraic spectrum.

Note that these options select the eigenvalue properties for eigenvalues of OP (and B for **Generalized** problems), the linear operator determined by the computational mode and problem type.

List Default $=$ **Nolist**
Nolist

Normally each optional parameter specification is not printed to the advisory channel as it is supplied. **List** may be used to enable printing and **Nolist** may be used to suppress the printing.

Monitoring i Default $= -1$

If $i > 0$, monitoring information is output to channel number i during the solution of each problem; this may be the same as the **Advisory** channel number. The type of information produced is dependent on the value of **Print Level**, see the description of **Print Level** in this section for details of the information produced. Please see X04ACF to associate a file with a given channel number.

Pointers Default $=$ No

During the iterative process and reverse communication calls to F12FBF, required data can be communicated to and from F12FBF in one of two ways. When **Pointers** $=$ No is selected (the default) then the array arguments X and MX are used to supply you with required data and used to return computed values back to F12FBF. For example, when $\mathrm{IREVCM} = 1$ F12FBF returns the vector x in X and the matrix-vector product Bx in MX and expects the result of the linear operation $\mathrm{OP}(x)$ to be returned in X.

If **Pointers** $=$ Yes is selected then the data is passed through sections of the array argument COMM. The section corresponding to X when **Pointers** $=$ No begins at a location given by the first element of ICOMM; similarly the section corresponding to MX begins at a location given by the second element of ICOMM. This option allows F12FBF to perform fewer copy operations on each intermediate exit and entry, but can also lead to less elegant code in the calling program.

Print Level i Default $= 0$

This controls the amount of printing produced by F12FDF as follows.

$= 0$ No output except error messages. If you want to suppress all output, set **Print Level** $= 0$.
≥ 0 The set of selected options.
$= 2$ Problem and timing statistics on final exit from F12FBF.
≥ 5 A single line of summary output at each Lanczos iteration.
≥ 10 If **Monitoring** > 0, then at each iteration, the length and additional steps of the current Lanczos factorization and the number of converged Ritz values; during re-orthogonalisation, the norm of initial/restarted starting vector; on a final Lanczos iteration, the number of update iterations taken, the number of converged eigenvalues, the converged eigenvalues and their Ritz estimates.

≥ 20 Problem and timing statistics on final exit from F12FBF. If **Monitoring** > 0, then at each iteration, the number of shifts being applied, the eigenvalues and estimates of the symmetric tridiagonal matrix H, the size of the Lanczos basis, the wanted Ritz values and associated Ritz estimates and the shifts applied; vector norms prior to and following re-orthogonalisation.

≥ 30 If **Monitoring** > 0, then on final iteration, the norm of the residual; when computing the Schur form, the eigenvalues and Ritz estimates both before and after sorting; for each iteration, the norm of residual for compressed factorization and the symmetric tridiagonal matrix H; during re-orthogonalisation, the initial/restarted starting vector; during the Lanczos iteration loop, a restart is flagged and the number of the residual requiring iterative refinement; while applying shifts, some indices.

≥ 40 If **Monitoring** > 0, then during the Lanczos iteration loop, the Lanczos vector number and norm of the current residual; while applying shifts, key measures of progress and the order of H; while computing eigenvalues of H, the last rows of the Schur and eigenvector matrices; when computing implicit shifts, the eigenvalues and Ritz estimates of H.

≥ 50 During Lanczos iteration loop: norms of key components and the active column of H, norms of residuals during iterative refinement, the final symmetric tridiagonal matrix H; while applying shifts: number of shifts, shift values, block indices, updated tridiagonal matrix H; while computing eigenvalues of H: the diagonals of H, the computed eigenvalues and Ritz estimates.

Note that setting **Print Level** ≥ 30 can result in very lengthy **Monitoring** output.

Random Residual Default = **Random Residual**
Initial Residual

To begin the Lanczos iterative process, F12FBF requires an initial residual vector. By default F12FBF provides its own random initial residual vector; this option can also be set using **Random Residual**. Alternatively, you can supply an initial residual vector (perhaps from a previous computation) to F12FBF through the array argument RESID; this option can be set using **Random Residual**.

Regular Default = **Regular**
Regular Inverse
Shifted Inverse
Buckling
Cayley

These options define the computational mode which in turn defines the form of operation OP(x) to be performed when F12FBF returns with IREVCM $= -1$ or IREVCM $= 1$ and the matrix-vector product Bx when F12FBF returns with IREVCM $= 2$.

Given a **Standard** eigenvalue problem in the form $Ax = \lambda x$ then the following modes are available with the appropriate operator OP(x).

Regular OP $= A$
Shifted Inverse OP $= (A - \sigma I)^{-1}$ where σ is real

Given a **Generalized** eigenvalue problem in the form $Ax = \lambda Bx$ then the following modes are available with the appropriate operator OP(x).

Regular Inverse OP $= B^{-1}A$
Shifted Inverse OP $= (A - \sigma B)^{-1}B$, where σ is real
Buckling OP $= (B - \sigma A)^{-1}A$, where σ is real
Cayley OP $= (A - \sigma B)^{-1}(A + \sigma B)$, where σ is real

Standard Default = **Standard**
Generalized

The problem to be solved is either a standard eigenvalue problem, $Ax = \lambda x$, or a generalized eigenvalue problem, $Ax = \lambda Bx$. The option **Standard** should be used when a standard eigenvalue problem is being solved and the option **Generalized** should be used when a generalized eigenvalue problem is being solved.

Tolerance r Default $= \epsilon$

An approximate eigenvalue has deemed to have converged when the corresponding Ritz estimate is within **Tolerance** relative to the magnitude of the eigenvalue.

Vectors Default = Schur

The routine F12FCF can optionally compute the Schur vectors and/or the eigenvectors corresponding to the converged eigenvalues. To turn off computation of any vectors the option **Vectors** = None should be set. To compute only the Schur vectors (at very little extra cost), the option **Vectors** = Schur should be set and these will be returned in the array argument V of F12FCF. To compute the eigenvectors (Ritz vectors) corresponding to the eigenvalue estimates, the option **Vectors** = Ritz should be set and these will be returned in the array argument Z of F12FCF; if the array argument V is passed to F12FCF in place of Z then the Schur vectors in V are overwritten by the eigenvectors computed by F12FCF.
